

A Generic Deriving Mechanism for Haskell

José Pedro Magalhães
Atze Dijkstra, Johan Jeuring, Andres Löh

Haskell SG Meetup
Singapore

June 7, 2018

Outline

Overview

Viewpoints

End user

Compiler implementer

Library writer

Conclusion

Overview

- ▶ Haskell has a number of (built-in) type classes that can automatically be derived: **Bounded**, **Enum**, **Eq**, **Ord**, **Read**, and **Show**
- ▶ This talk is about a mechanism that lets you define these classes and your own *in* Haskell such that they can be derived automatically
- ▶ Implemented in the Glasgow Haskell Compiler

Features

We can:

- ▶ Handle meta-information such as constructor names and field labels
- ▶ Derive all the Haskell 98 classes
- ▶ Derive most of the classes that GHC can derive, including **Typeable** and classes of kind $\star \rightarrow \star$ such as **Functor**

Outline

Overview

Viewpoints

End user

Compiler implementer

Library writer

Conclusion

Using generic functions

If a class is generic, it can be used in a **deriving** construct.

Assuming a type class

```
data Bit = 0 | 1
class Encode  $\alpha$  where
  encode ::  $\alpha \rightarrow$  [Bit]
```

The end user can write

```
data Exp = Const Int | Plus Exp Exp
deriving (Show, Encode)
```

and then use

```
test :: [Bit]
test = encode (Plus (Const 1) (Const 2))
```

Outline

Overview

Viewpoints

End user

Compiler implementer

Library writer

Conclusion

Basic idea

- ▶ For each datatype, there is an equivalent internal representation.
- ▶ All the concepts contained in the **data** construct (application, abstraction, choice, sequence, recursion) are captured by a limited set of *representation types*.
- ▶ The compiler generates an internal representation for every datatype, together with conversion functions and derived instances

Type representation

- ▶ The type representation is available in a module (`Generics.Deriving.Base`).
- ▶ The representation types need to be bundled with the compiler (much like `Data.Data` for `syb` on GHC), but the library itself (`generic-deriving` on Hackage) is portable.
- ▶ The library contains a set of datatypes as well as a class that allows conversion between a datatype and its representation.

Example

```
data Exp = Const Int | Plus Exp Exp
```

```
type Rep0Exp =  
  D1 $Exp ( C1 $ConstExp (Rec0 Int)  
            + C1 $PlusExp (Rec0 Exp × Rec0 Exp))
```

Example

```
data Exp = Const Int | Plus Exp Exp
```

```
type RepExp0 =  
    ( Int )  
+ ( Exp × Exp )
```

Note that the representation is *shallow* – recursive calls are to `Exp`, not `RepExp0`.

Most of the representation is meta-information about:

Example

```
data Exp = Const Int | Plus Exp Exp
```

```
type Rep0Exp =  
  D1 $Exp (      (      Int)  
            +      (      Exp ×      Exp))
```

Note that the representation is *shallow* – recursive calls are to `Exp`, not `Rep0Exp`.

Most of the representation is meta-information about:

- ▶ the datatype itself,

Example

```
data Exp = Const Int | Plus Exp Exp
```

```
type Rep0Exp =  
  D1 $Exp ( C1 $ConstExp ( Int)  
            + C1 $PlusExp ( Exp × Exp))
```

Note that the representation is *shallow* – recursive calls are to `Exp`, not `Rep0Exp`.

Most of the representation is meta-information about:

- ▶ the datatype itself,
- ▶ the constructors,

Example

```
data Exp = Const Int | Plus Exp Exp
```

```
type Rep0Exp =  
  D1 $Exp ( C1 $ConstExp (Rec0 Int)  
            + C1 $PlusExp (Rec0 Exp × Rec0 Exp))
```

Note that the representation is *shallow* – recursive calls are to `Exp`, not `Rep0Exp`.

Most of the representation is meta-information about:

- ▶ the datatype itself,
- ▶ the constructors,
- ▶ where recursive calls take place.

Lifting

Our approach can handle type classes with parameters of both

- ▶ kind \star such as **Encode** and **Show**;
- ▶ kind $\star \rightarrow \star$ such as **Functor**.

We therefore represent *all* datatypes at kind $\star \rightarrow \star$.

Types of kind \star get a dummy parameter in their representation.

Representation types

data V_1 ρ

data U_1 $\rho = U_1$

data $(+)$ $\phi \psi \rho = L_1 (\phi \rho) \mid R_1 (\psi \rho)$

data (\times) $\phi \psi \rho = \phi \rho \times \psi \rho$

The void type V_1 is for types without constructors.

The unit type U_1 is for constructors without fields.

Sums represent choice between constructors.

Products represent sequencing of fields.

Meta-information

data $K_1 \iota \gamma \quad \rho = K_1 \gamma$

data $M_1 \iota \mu \phi \rho = M_1 (\phi \rho)$

These types record additional information, such as names and fixity, for instance. They are instantiated as follows:

data D -- datatypes

data C -- constructors

data S -- record selectors

data R -- recursive calls

data P -- parameters

type $D_1 = M_1 D$

type $C_1 = M_1 C$

type $S_1 = M_1 S$

type $Rec_0 = K_1 R$

type $Par_0 = K_1 P$

We group five combinators into two because we often do not care about all the different types of meta-information.

Example: meta-information for expressions

GHC automatically generates the following for `Exp`:

```
data $Exp
data $ConstExp
data $PlusExp
instance Datatype $Exp where
  moduleName _ = "ModuleName"
  datatypeName _ = "Exp"
instance Constructor $ConstExp where conName _ = "Const"
instance Constructor $PlusExp where conName _ = "Plus"
```

The classes `Datatype` and `Constructor` can hold more information if desired.

Conversion

We use a type class to mediate between values and representations:

```
class Generic  $\alpha$  where  
  type Rep  $\alpha :: \star \rightarrow \star$   
  from  $:: \alpha \rightarrow$  Rep  $\alpha \chi$   
  to    $::$  Rep  $\alpha \chi \rightarrow \alpha$ 
```

Conversion

We use a type class to mediate between values and representations:

```
class Generic  $\alpha$  where  
  type Rep  $\alpha :: \star \rightarrow \star$   
  from  $:: \alpha \rightarrow$  Rep  $\alpha \chi$   
  to    $::$  Rep  $\alpha \chi \rightarrow \alpha$ 
```

Instance for `Exp` (automatically generated by GHC):

```
instance Generic Exp where  
  type Rep Exp = Rep0Exp  
  from (Const n) = M1 (L1 (M1 (K1 n)))  
  from (Plus e e') = M1 (R1 (M1 (K1 e × K1 e')))  
  to (M1 (L1 (M1 (K1 n)))) = Const n  
  to (M1 (R1 (M1 (K1 e × K1 e')))) = Plus e e'
```

Compiler support

For each datatype, the compiler generates the following:

- ▶ Meta-information, i.e. datatypes and class instances.
- ▶ Representation type synonym(s).
- ▶ **Generic** and/or **Generic₁** instance.

Each **deriving** construct simply gives rise to an empty instance (more on that later).

Design choices

There is a certain amount of flexibility in how the compiler generates the representation.

For example, sums and products are currently balanced.

It is not clear how much of these details should be part of the specification.

Outline

Overview

Viewpoints

End user

Compiler implementer

Library writer

Conclusion

Generic function definitions

The library writer defines generic (derivable) functions.
We use two classes: one for the base types (kind \star):

```
class Encode  $\alpha$  where  
  encode ::  $\alpha \rightarrow [\text{Bit}]$ 
```

and one for the representation types (kind $\star \rightarrow \star$):

```
class Encode1  $\phi$  where  
  encode1 ::  $\phi \chi \rightarrow [\text{Bit}]$ 
```


Simple cases

The generic cases are defined as instances of `Encode1`:

instance `Encode1 V1 where`

`encode1 _ = []`

instance `Encode1 U1 where`

`encode1 _ = []`

instance `(Encode1 ϕ) \Rightarrow Encode1 (M1 ι γ ϕ) where`

`encode1 (M1 a) = encode1 a`

Sums and products

instance (Encode₁ ϕ , Encode₁ ψ) \Rightarrow Encode₁ ($\phi + \psi$) **where**
 encode₁ (L₁ a) = 0 : encode₁ a
 encode₁ (R₁ a) = 1 : encode₁ a

instance (Encode₁ ϕ , Encode₁ ψ) \Rightarrow Encode₁ ($\phi \times \psi$) **where**
 encode₁ (a \times b) = encode₁ a $\#$ encode₁ b

Constants and base types

For constants, we rely on `Encode`:

```
instance (Encode  $\alpha$ )  $\Rightarrow$  Encode1 (K1  $\iota$   $\alpha$ ) where  
  encode1 (K1 a) = encode a
```

In this way we close the recursive loop: if α is a representable type, `encode` will call `from` and then `encode1` again.

For base types, we need to provide ad-hoc instances:

```
instance Encode Int where encode = ...  
instance Encode Char where encode = ...
```

Default generic instance

The generic case is provided by generic defaults:

```
class Encode  $\alpha$  where  
  encode      ::  $\alpha \rightarrow [\text{Bit}]$   
  default encode :: (Generic  $\alpha$ , Encode1 (Rep  $\alpha$ ))  
                   $\Rightarrow \alpha \rightarrow [\text{Bit}]$   
  encode x = encode1 (from x)
```

These are just like regular default methods, only with a different type signature.

Using the generic instance

We are done:

```
data Exp = Const Int | Plus Exp Exp deriving Encode
```

will cause the generation of

```
instance Encode Exp where  
  encode x = encode1 (from x)
```

Back to the internals: kind $\star \rightarrow \star$ types

For type constructors (kind $\star \rightarrow \star$) we use a few more representation types:

newtype Par_1 $\rho = \text{Par}_1 \rho$

newtype $\text{Rec}_1 \phi \rho = \text{Rec}_1 (\phi \rho)$

newtype $(\circ) \phi \psi \rho = \text{Comp}_1 (\phi (\psi \rho))$

We use Par_1 to store the parameter, Rec_1 to encode recursive occurrences of type constructors, and \circ for type composition (eg. lists of trees).

Example: representing lists I

```
data List  $\rho$  = Nil | Cons  $\rho$  (List  $\rho$ )  
deriving (Show, Encode, Functor)
```

The compiler generates instance of **Generic** for kind \star functions:

```
type Rep0List  $\rho$  =  
  D1 $List ( C1 $NilList U1  
             + C1 $ConsList (Par0  $\rho$  × Rec0 (List  $\rho$ )))
```

```
instance Generic (List  $\rho$ ) where
```

```
type Rep (List  $\rho$ ) = Rep0List  $\rho$   
from Nil           = M1 (L1 (M1 U1))  
from (Cons h t)   = M1 (R1 (M1 (K1 h × K1 t)))  
to (M1 (L1 (M1 U1)))           = Nil  
to (M1 (R1 (M1 (K1 h × K1 t)))) = Cons h t
```

Example: representing lists II

```
type Rep0List ρ =  
  D1 $List ( C1 $NilList U1  
             + C1 $ConsList (Par0 ρ × Rec0 (List ρ)))
```

And an instance of **Generic₁** for kind $\star \rightarrow \star$ functions:

```
type Rep1List = D1 $List ( C1 $NilList U1  
                             + C1 $ConsList (Par1 × Rec1 List))
```

instance Generic₁ List **where**

```
type Rep1 List = Rep1List  
from1 Nil           = M1 (L1 (M1 U1))  
from1 (Cons h t) = M1 (R1 (M1 (Par1 h × Rec1 t)))  
to1 (M1 (L1 (M1 U1)))           = Nil  
to1 (M1 (R1 (M1 (Par1 h × Rec1 t)))) = Cons h t
```


Back to the library writer: generic map I

We show how to define **Functor** generically as an example of a kind $\star \rightarrow \star$ function. For consistency, we again use two type classes:

class Functor ϕ **where**

fmap :: $(\rho \rightarrow \alpha) \rightarrow \phi \rho \rightarrow \phi \alpha$

default fmap :: (Generic₁ ϕ , Functor₁ (Rep₁ ϕ))

$\Rightarrow (\rho \rightarrow \alpha) \rightarrow \phi \rho \rightarrow \phi \alpha$

fmap f x = to₁ (fmap₁ f (from₁ x))

class Functor₁ ϕ **where**

fmap₁ :: $(\rho \rightarrow \alpha) \rightarrow \phi \rho \rightarrow \phi \alpha$

Generic map II

Most cases are trivial:

instance Functor₁ U₁ **where**

$$\text{fmap}_1 f U_1 = U_1$$

instance Functor₁ (K₁ ι γ) **where**

$$\text{fmap}_1 f (K_1 a) = K_1 a$$

instance (Functor₁ ϕ) \Rightarrow Functor₁ (M₁ ι γ ϕ) **where**

$$\text{fmap}_1 f (M_1 a) = M_1 (\text{fmap}_1 f a)$$

instance (Functor₁ ϕ , Functor₁ ψ) \Rightarrow Functor₁ ($\phi + \psi$) **where**

$$\text{fmap}_1 f (L_1 a) = L_1 (\text{fmap}_1 f a)$$

$$\text{fmap}_1 f (R_1 a) = R_1 (\text{fmap}_1 f a)$$

instance (Functor₁ ϕ , Functor₁ ψ) \Rightarrow Functor₁ ($\phi \times \psi$) **where**

$$\text{fmap}_1 f (a \times b) = \text{fmap}_1 f a \times \text{fmap}_1 f b$$

Generic map II

The most interesting instance is the one for parameters:

```
instance Functor1 Par1 where  
  fmap1 f (Par1 a) = Par1 (f a)
```

Recursion and composition rely on **Functor**:

```
instance (Functor  $\phi$ )  $\Rightarrow$  Functor1 (Rec1  $\phi$ ) where  
  fmap1 f (Rec1 a) = Rec1 (fmap f a)
```

```
instance (Functor  $\phi$ , Functor1  $\psi$ )  $\Rightarrow$  Functor1 ( $\phi \circ \psi$ ) where  
  fmap1 f (Comp1 x) = Comp1 (fmap (fmap1 f) x)
```

Generic map III

Now the compiler can **derive Functor** for **List**:

```
instance Functor List where  
  fmap f x = to1 (fmap1 f (from1 x))
```

Outline

Overview

Viewpoints

End user

Compiler implementer

Library writer

Conclusion

Conclusion

- ▶ The deriving mechanism does not have to be “magic”: it can be explained in Haskell.
- ▶ Derivable functions become accessible and portable.
- ▶ We provide an implementation in GHC and detailed information on how to implement it for other compilers.
- ▶ We hope that the behaviour of derived instances can be redefined in Haskell Prime, perhaps along the lines of our work.